


Lesson 12: a Laser Beam Shaper, with Aspherics

In lesson 11 we designed a laser beam shaper to flatten out the Gaussian beam profile of a small HeNe laser. To keep manufacturing costs down, we tried in that lesson to do the job with spherical surfaces, since those are much easier to make than aspherics. We came up with a six-element design that seemed to meet our specs. Perhaps that design could be further improved, but we also must ask whether six spherical lenses will be cheaper to make than two aspherics. If not, then an aspheric design looks more attractive.

Let us start with the same two-element configuration that we used in Lesson 11, modified so that we only flatten the flux out to the $1/e^{**2}$ point. Going out to twice that aperture appeared to be impractical before, and we suspect that it will again. Here is our starting point):

```
RLE                                ! Beginning of lens input file.
ID LASER BEAM SHAPER
WA1 .6328                          ! Single wavelength
UNI MM                              ! Lens is in millimeters
OBG .35 1                          ! Gaussian object; waist radius -.35 mm; define full aperture at the 1/e**2 point.
1 TH 22                             ! Surface 2 is 22 mm from the waist .
2 RD -5 TH 2 GTB S                  ! Guess some reasonable lens parameters; use glass type SF6 from Schott catalog
SF6
3 UMC 0.3 YMT 5                    ! Solve for the curvature of surface 3 so the marginal ray has an angle of 0.3; find
!spacing so ray height is 5 mm on next surface
4 RD 20 TH 4 PIN 2                 ! Guesses for surface 4
5 UMC 0 TH 50                      ! Solve for curvature of 5 so beam is collimated.
7                                  ! Surfaces 6 and 7 exist
AFOCAL                             ! because they are required for AFOCAL output.
END                                 ! End of lens input file.
```



Since we learned a great deal in Lesson 11, we can start this one with this merit function..

```
CHG
NOP                                ! Be sure there are no pickups or solves.
4 PIN 2
5 TH 10 UMC 0                      ! move surface 6 before the caustic
END

PANT                                ! Start of variable parameter definition.
VLIST RAD 2 3 4 5                  ! Vary four radii.
VLIST TH 3                          ! Vary the central airspace.
VY 3 CC                              ! Vary the conic constant on surface 3.
VY 4 CC                              ! And on surface 4.
VY 3 G 3                            ! Add three aspheric terms to surface 3.
VY 3 G 6
```

```

VY 3 G 10
VY 4 G 3          ! And three to surface 4.
VY 4 G 6
VY 4 G 10
END

AANT              ! Start of merit function definition.
AEC               ! Enable automatic edge feathering control.
ACC              ! Enable automatic center thickness monitoring
ASC              ! Enable automatic slope control, so curves don't get too steep.
LUL 100 1 1 A TOTL      ! Limit the paraxial total length to no more than 150 mm.
M 5 100 A P YA 0 0 1 0 LB1
M 5 100 A P YA 0 0 1 0 LB2 ! Assign a target of 5 mm to the marginal ray on surfaces 5, 6.

M 0 1 A P FLUX 0 0 1 0 LB1 ! Target the flux difference between the marginal ray point and the on-
!axis point to 0 on surface 6.
M 0 1 A P FLUX 0 0 .99 0 LB1 ! Target the flux at the 0.99 aperture point.
M 0 1 A P FLUX 0 0 .98 0 LB1 ! And so on, for a set of zones.
M 0 1 A P FLUX 0 0 .97 0 LB1
M 0 1 A P FLUX 0 0 .96 0 LB1
M 0 1 A P FLUX 0 0 .95 0 LB1
M 0 1 A P FLUX 0 0 .94 0 LB1
M 0 1 A P FLUX 0 0 .93 0 LB1
M 0 1 A P FLUX 0 0 .92 0 LB1
M 0 1 A P FLUX 0 0 .91 0 LB1
M 0 1 A P FLUX 0 0 .9 0 LB1
M 0 1 A P FLUX 0 0 .89 0 LB1
M 0 1 A P FLUX 0 0 .88 0 LB1
M 0 1 A P FLUX 0 0 .86 0 LB1
M 0 1 A P FLUX 0 0 .84 0 LB1
M 0 1 A P FLUX 0 0 .82 0 LB1
M 0 1 A P FLUX 0 0 .8 0 LB1
M 0 1 A P FLUX 0 0 .7 0 LB1
M 0 1 A P FLUX 0 0 .5 0 LB1
M 0 1 A P FLUX 0 0 .3 0 LB1
GSO 0 .01 10 P      ! Target the OPD of an SFAN of 10 rays to zero, with a weight of .01
GSR 0 50 10 P      ! And also target the ray angles to zero.
END

SNAP
SYNO 50

```


While this is pretty straightforward, we should point out some things: Why is GSR used to target the ray angles? Usually, GSR controls the actual X-coordinate of each ray relative to the chief ray – but since this system is in AFOCAL mode, where the output is collimated, this entry targets the output *angles* instead.

Where did we specify that the ray and flux targets are to be taken on surface 6? Well, this system has a total of seven surfaces, counting the two dummies at the end that are required for AFOCAL angle conversion. The mnemonic “LB1” means “last but one”, and it gets replaced by the number 6 when the input is processed.


Here, we have chosen to vary the conic constant and three aspheric terms on two surfaces. There are also higher-order terms available, using a variety of surface specifications – but this is a simple task and we want to see how close we can come with just those terms. This form of aspheric has 22 terms available, but only terms G 3, G 6, G 10, G 16, G 18, G 19, G 20, G 21, and G 22 are rotationally symmetric. They vary the 4th, 6th, 8th, and 10th - through 20th-order aspheric terms, and here we did not even use the last six.

Okay, let's run this MACro. Things might still get better, so we anneal for a few cycles.

That brings the merit function down to 2.1E-5, which is a sign that we have reached a good solution.



Here is the FLUX plot for our final design:



Wow! The flux is almost perfectly uniform! How about the OPD errors?

SYNOPTSYS AI>OPD

SYNOPTSYS AI>SFA 5 P

ID LASER BEAM SHAPER 115 20-MAY-17 13:32:54
SAGITTAL RAY FAN ANALYSIS

FRACT. OBJECT HEIGHT HBAR 0.000000 GBAR 0.000000
COLOR NUMBER 1

REL ENT PUPIL XEN	WAVEFRONT OPD (WAVES)	ABERR
0.200	-0.000865	
0.400	-6.675373E-05	
0.600	0.000361	
0.800	-0.000651	


1.000

-0.000791

Okay, this design is essentially perfect, with less than 1/1000 wave of error, and it only required two elements. We can be pleased at that. Looks like we don't need the six-element design of Lesson 11 after all!

Just to be sure, let's also examine the output wavefront with DPROP:

```
STORE 9
CHG
1 SIN
1 TH 0
CFIX
END
DPROP P 0 0 6 SURF 2.5 R RESAMPLE
GET 9
```





Yes, this is exactly what we were after. Here again, the program has determined that diffraction does not play a significant role, and the calculation is purely geometric.

Now the only question is how difficult the aspherics will be to manufacture. To see how far the aspherics are from the closest fitting sphere (CFS), we enter


```
ADEF 3 PLOT
```

```
ADEF 4 PLOT
```



Both of these aspherics are only a few microns away from the CFS. It appears that this is manageable. Let us look at the fringe pattern, relative to the CFS:

ADEF 3 FRINGES



This may or may not be a challenge to the folks in the shop. Depending on how the aspherics are made and measured, one might try to reduce the aspheric departure somewhat, trading off that departure against the performance. See Lesson 21 to see how one can use the CLINK optimization feature to do just that.

